


DRAGONFLY

Assembly Instructions



Introduction

- Six Digit Seven Segment bright HE Red display with decimal point*
- Six pushbuttons, variable resistor, 32768Hz crystal time-base & RS485 communications
- General purpose USER I/O port including hardware I2C and A/D, 1 wire or TIMER0 connector
- PIC16F886 (default) or PIC18F2525 (optional) microcontroller compatible
- Requires an ICD type programmer / debugger such as Junebug or Inchworm (not included)

Dragonfly was designed as an introduction to working with multiplexed displays and scanning keypads, it also has many practical uses from the obvious digital clock (see program at the end of this document) to countless other experiments including test instruments. Some possible projects might be...

Biorhythm clock, Marque display, Thermostat, Thermometer, iButton™ reader, Lottery picker, Reaction tester, Dual 3 digit voltmeter, I2C tutor, RS485 node, Communicating clock, GPS clock / display, 50MHz Frequency counter, Stopwatch, Timer, Toothbrush timer... The possibilities are endless as the device can be reprogrammed in seconds.

Dragonfly has full debug support with both Junebug and Inchworm kits.

If you're documenting your project download the free seven segment font `myblueroom.ttf` from <http://www.blueromelectronics.com>


HELLO dragonFLY 12:36 22°C

* Any or all seven segment displays can be inverted to form colons and degree symbols at assembly time

Necessary Tools *(not included in kit)*

As with any electronic kit the following tools are essential:

- Low wattage fine tip soldering iron <50W
- Resin core solder
- Wire cutters or side cutters (small)
- Needle nose pliers (small)
- Slotted screwdriver (small)
- Phillips screwdriver (small)
- Multimeter (this really is a must for any electronics project)



Before You Build...

It's possible to cut down standard IC sockets for the displays so they can be inverted at a later date. But if you decide to solder the displays the projects presented will always be for the setup as used on the cover of this manual. It has digits 1,2,4 & 5 with the decimal point at the bottom and digits 3 & 6 with the decimal point inverted (upper left). This version allows for a colon between digits 2 & 3 (*12.34*) and two degree pairs for temperature display (*23c28c*) and is the default *recommended* setup. It's possible to use non HE or Super Red displays but R1 thru R8 will have to be adjusted to brighten up less efficient displays, the low power HE display can also be powered from the Junebugs USB supply.

Assembly

Traditionally it's easiest to assemble a circuit board with the lowest profile and/or smallest parts first.

Install all resistors R1 thru R19

Small signal diodes D1 thru D7 (Red 1N4148)

Note: diodes use a colored band to denote polarity

IC sockets U1, U2 and U3 (*notice notch orientation*)

Pushbuttons SW1 thru SW6

Crystal Y1 (*not too snug or use an insulator*)

Transistors Q1 thru Q6 (NPN 2N2222)

Capacitors C3, C4, C5 (*note lead spacing*)

Capacitors C1, C2 (68pf)


Capacitor C6 (*note polarity*) (47uF)

Trimmer potentiometer VR1

Connectors CON1 thru CON4

Displays DISP1 thru DISP6

(Notice DISP3 and DISP6 are installed upside-down this is optional but allows a colon or degree symbol)



Final Assembly

Install U2 (CD4022), U1 (16F886 or 18F2525) and optionally U3 (SN75176)

Initial Testing


You can program Dragonfly with the HH:MM:SS clock program starting on page 5, this is a good program get acquainted with some of the many features on the kit.



Parts List Dragonfly

Capacitors

2	C1, 2	12pf (may vary see 32.768KHz crystal spec)
3	C3, 4, 5	0.1uf
1	C6	47uf 6.3V



Resistors 1/4W 5% Carbon


1	R10	330	Orange, Orange, Brown, Gold
6*	R1, 2, 3, 4, 5, 6, 7, 8	470	Yellow, Violet, Brown, Gold
6	R12, 13, 14, 15, 16, 17	1.2K	Brown, Red, Red, Gold
1	R11	4.7K	Yellow, Violet, Red, Gold
3	R9, 18, 19	22K	Red, Red, Orange, Gold
1	VR1	10K	Trimmer Potentiometer

Semiconductors

6	DISP1, 2, 3, 4, 5, 6	14mm (0.56") Seven Segment Common Cathode Super Red Display(HE)
7	D1, 2, 3, 4, 5, 6, 7	1N4148 Diode
1	LED1	3mm Red LED
6	Q1, 2, 3, 4, 5, 6	2N2222 or 2N4401 NPN Transistor
1	U1	PIC16F886 or PIC18F2525
1	U2	CD4022 CMOS Ten stage Johnson counter

Crystal

1	Y1	32.768KHz crystal 12.5pf
---	----	--------------------------



Connectors

1	CON1	ICD 2x5 PCB Male
1	CON2	8 pin female inline socket (see text)
1	CON3	8 pin male plug (low profile) USER
1	CON4	3 pin male plug (low profile) RS485
1	CON5	3 pin machine socket
1	CON6	R/A USB A or Mini B connector digikey# WM17115-ND

Switches

6	SW1, 2, 3, 4, 5, 6	small pushbutton switch
---	--------------------	-------------------------


Miscellaneous

1	28-pin	DIP IC Socket 0.3"
1	8-pin	DIP IC Socket 0.3"

Options and Accessories

1	Hammond 1591B	Enclosure (either top or bottom will work)
1	U3	SN75176 RS485 transceiver (or MAX485)

**it's also possible to use a single 470ohm x 8 resistor DIP*



http://www.blueroomelectronics.com	
Title	Dragonfly Digital Display & RS485
Size	A
Document Number	wrichardson@blueroomelectronics.com
Rev	B
Date:	Sunday, September 16, 2007
Sheet	1 of 1

Dragonfly HH:MM.SS Clock (24hr)

12:34.56

The Dragonfly clock uses a 32768Hz watch crystal and inverts is designed for the suggested digit configuration of 3 & 6 inverted.

```
; ** Dragonfly Digital Clock HH:MM.SS with flashing colon
; ** Button (1) Hours+1, (3) Minutes+1, (5) Seconds+1
; ** Button (6) resets Seconds to zero & holds clock till released
    list    p=16F886
    include <p16F886.inc>
    __CONFIG __CONFIG1, _DEBUG_ON & _WDT_OFF & _LVP_OFF & _INTOSC10
    errorlevel    -302,-305
    cblock    0x20    ; \
    Buttons    ; flags for pushbuttons 1 = pressed
    Digit:6
    Jiffy    ; 1/256 second timebase
    HalfSec    ; .5 second + seconds
    Minutes    ;
    Hours    ; hours
    DigSel    ; which digit 0-5 (left to right)
    Tens    ; 10s holder for display
    Ones    ; 1s holder for display
    Endc    ; /
#define Rst4022 PORTC, 5    ; CD4022 reset active low
#define Clk4022 PORTC, 2    ; CD4022 clock low to high increments
#define PB_In    PORTA, 3    ; push buttons 1 thru 6
    org    0    ; reset vector
    nop
    goto    Init
    org    4    ; interrupt vector (does not save W, STATUS, etc)
    goto    _512    ; jump to interrupt service routine 512
_Seg    addwf    PCL    ; 7-segment display table
    dt    b'00111111'    ; "0"
    dt    b'00000110'    ; "1"
    dt    b'10011011'    ; "2"
    dt    b'10001111'    ; "3"
    dt    b'10100110'    ; "4"
    dt    b'10101101'    ; "5"
    dt    b'10111101'    ; "6"
    dt    b'00000111'    ; "7"
    dt    b'10111111'    ; "8"
    dt    b'10101111'    ; "9"
_PB    addwf    PCL    ; byte to bit (1-6) table
    dt    0x02, 0x04, 0x08, 0x10, 0x20, 0x40
_512    incfsz    Jiffy    ; 1/512 second refresh clock
    goto    _DoDisp    ; refresh display only
    rrf    HalfSec, W    ; rotate 0.5 second bit to carry
    bnc    _Sec    ; don't update keypad
_KeyPad    movfw    Buttons
    bz    _Sec    ; nothing to update
    bsf    PORTA, 5    ; clear the power loss indicator
    btfsc    Buttons, 3    ; button 3 pressed?
    goto    _Min    ; increment 1 Minute
    btfsc    Buttons, 1    ; button 1 pressed?
    goto    _Hrs    ; increment 1 Hour
    btfsc    Buttons, 6    ; button 6 pressed?
    clrf    HalfSec    ; reset seconds and hold
    btfss    Buttons, 5    ; button 5 pressed?
    goto    _Sec
    bsf    HalfSec, 0    ; set the 1/2 second count
```

```

_Sec    incf    HalfSec
        incf    HalfSec          ; .5 second increments
        movlw   .120
        subwf   HalfSec,w
        bnc     _UpSec
        clrf    HalfSec          ; zero the half seconds
_Min    incf    Minutes          ; 0-59 minutes
        movlw   .60
        subwf   Minutes,w
        bnc     _UpMin
        clrf    Minutes
_Hrs    incf    Hours
        movlw   .24              ; 0-23 hours
        subwf   Hours,w
        skpnc
        clrf    Hours          ; reset hours
_UpHrs  movlw   Digit          ; point to hours display (ones)
        addlw   .1
        movwf   FSR
        incf    Hours,w        ; 1-24 hours
        call    _DecBCD        ; covert byte to ones & tens
        decf    Tens,w
        skpnz
        clrf    Digit          ; blank lead digit (tens of hours)
_UpMin  movlw   Digit          ; point to minutes display (ones)
        addlw   .3              ; add digit offset
        movwf   FSR
        movfw   Minutes
        call    _DecBCD        ; covert byte to ones & tens
        btfsc   Digit+2,7      ; digit 3 dp / 6 transpose
        bsf     Digit+2,6
        bcf     Digit+2,7
_UpSec  movlw   Digit          ; point to seconds display (ones)
        addlw   .5              ; add digit offset
        movwf   FSR
        clrc     ; make sure carry is clear
        rrf     HalfSec,W
        call    _DecBCD        ; covert byte to ones & tens
        btfsc   Digit+5,7      ; digit 6 dp / 6 transpose
        bsf     Digit+5,6
        bcf     Digit+5,7
_Colon  bsf     Digit+3,6      ; turn on digit 4 colon
        bcf     Digit+1,6
        bcf     Digit+2,7
        btfsc   HalfSec,0
        goto    _DoDisp
        bsf     Digit+1,6      ; light colon (dp segment)
        bsf     Digit+2,7      ; light colon
;*** refresh display (Digit) and scan keyboard (Buttons)
_DoDisp movfw   DigSel
        call    _PB            ; convert byte to bit
        iorwf   Buttons,f      ; set to pressed
        btfsc   PB_In          ; button is not pressed?
        xorwf   Buttons,f      ; clear the button
        clrf    PORTB          ; blank display ABCDEF
        movlw   b'00111111'    ; mask off lower six bits
        andwf   PORTA          ; blank display dp G
        incf    DigSel          ; increment to 6 then reset
        movlw   .6              ; W = 6 (number of digits)
        subwf   DigSel,w        ; is DigSel < 6
        bnc     _disp          ; yes then proceed normally
_reset  bsf     Rst4022         ; no then reset the 4022 counter (1)
        clrf    DigSel          ; clear the display digit count
        bcf     Rst4022         ; enable the 4022 counter

```



```

_disp    bsf      Clk4022          ; increment CD4022 counter (rising edge)
         movlw    Digit           ; get pointer to Digit
         addwf    DigSel,w         ; add offset (pointer)
         movwf    FSR              ; point to display charater
         movfw    INDF             ; move charater to W
         movwf    PORTB           ; move W to PORTB
         andlw    b'11000000'     ; mask off low six bits
         iorwf    PORTA           ; segment G on (normal)
         bcf      Clk4022         ; clock low (does not incremnt clock)
         bcf      PIR1,CCP1IF     ; clear the interrupt flag
         retfie

;*** Decimal to BCD enter with byte 0-99 returns Ones
_DecBCD  movwf    Ones
         clrf     Tens
_10s     incf     Tens
         movlw    .10             ; load the constant 10
         subwf    Ones,f          ; subtract 10 and save result
         bc       _10s            ; check if answer negative
         movlw    .10
         addwf    Ones,w          ; W = Ones + 10
         call     _Seg            ; convert to seven segment
         movwf    INDF            ; save ones to display
         decf     FSR             ; next digit Tens (left)
         decf     Tens,W          ; W = Tens -1
         call     _Seg            ; convert to seven segment
         movwf    INDF            ; save tens to display
         return

;*** initialize the Dragonfly after a reset
Init     movlw    b'00001011'     ; timer1 osc on, sync,
         movwf    T1CON           ; enable external OSC on TMR1
         movwf    CCP1CON         ; special CCP timer compare & reset
         movlw    low(.64-1)      ; 32768
         movwf    CCPR1L
         movlw    high(.64-1)
         movwf    CCPR1H
         banksel ANSEL            ; B3
         movlw    b'00000001'    ;
         movwf    ANSEL           ; B3 all ports digital except RA0
         banksel TRISA           ; B1
         movlw    b'00011111'    ; segments dp & G, LED RX (pwr fail)
         movwf    TRISA           ; B1 I/O direction PORTA
         movlw    b'11000000'    ; segments ABCDEF, ICD2 PGC/PGD
         movwf    TRISB           ; B1 I/O direction PORTB
         movlw    b'11011011'    ; CD4022 clock & reset
         movwf    TRISC           ; B1 I/O direction PORTC
         bsf      PIE1,CCP1IE     ; B1 enable CCP1 interupt
         bcf      STATUS, RP0     ; B0
         movlw    b'11111011'    ; clock low and reset high
         movwf    PORTC           ; 4022 clock low & Reset high
         clrf     DigSel          ;
         clrf     Buttons         ;
         bcf      PORTA,5         ; set the power loss indicator
         movlw    .120            ; cause clock to overflow on reset
         movwf    HalfSec         ; any value over 119 will overflow
         movwf    Minutes         ; any value over 59 will overflow
         movlw    .10             ; start the clock at 12:00:00
         movwf    Hours
         movlw    b'11000000'
         movwf    INTCON          ; Enable GIE PIE
         bcf      Rst4022         ; enable the 4022
Loop     goto     $              ; loop forever
END

```

Dragonfly and other bluroomelectronics kits are available at

Retail Sales



255 College St. Toronto


Ontario, Canada

Tel (416) 977-9258

Fax (416) 977-4700

creatronpart@hotmail.com

<http://www.creatroninc.com>



Dealer Sales & Technical Inquiries



4550 Dufferin St. Toronto

Ontario, Canada

Tel (416) 897-1962

sales@bluroomelectronics.com

<http://www.bluroomelectronics.com>

Info and all other inquiries info@bluroomelectronics.com

